Problem:
Score : 400 points

Problem Statement

There is a knight - the chess piece - at the origin (0, 0) of a two-dimensional grid.

When the knight is at the square (i, j), it can be moved to either (i+1,j+2) or (i+2, j+1).

In how many ways can the knight reach the square (X, Y)?

Find the number of ways modulo 10^9 + 7.

Constraints

1 \leq X \leq 10^6

1 \leq Y \leq 10^6

All values in input are integers.

Input

Input is given from Standard Input in the following format:

X Y

Output

Print the number of ways for the knight to reach (X, Y) from (0, 0), modulo 10^9 + 7.

Sample Input 1

3 3

Sample Output 1

2

There are two ways: (0,0) \to (1,2) \to (3,3) and (0,0) \to (2,1) \to (3,3).

Sample Input 2

2 2

Sample Output 2

0

The knight cannot reach (2,2).

Sample Input 3

999999 999999

Sample Output 3

151840682

Print the number of ways modulo 10^9 + 7.

Solution in Haskell (GHC 8.8.3) (judge verdict: Accepted)

import qualified Data.Text.Lazy as T
import qualified Data.Text.Lazy.IO as T
import qualified Data.Text.Lazy.Read as T
import qualified Data.Vector.Unboxed as VU

main :: IO ()
main = do
  let
    modulus = 1000000007 :: Int
    modInv :: Int -> Int -> Int -> Int
    modInv modulus maxArg = (invs VU.!) . (`mod` modulus)
      where
        invs :: VU.Vector Int
        invs = VU.constructN (succ maxArg `min` modulus) rec
          where
            rec v = case VU.length v of
              0 -> 0
              1 -> 1
              l -> let (q, r) = modulus `divMod` l in (modulus - (v VU.! r) * q) `mod` modulus
    modComb :: Int -> Int -> Int
    modComb n = go 1
      where
        go :: Int -> Int -> Int
        go c 0 = c
        go c k = go (c * (n - k') `mod` modulus * modInv' k `mod` modulus) k'
          where
            k' = k - 1 :: Int
        modInv' :: Int -> Int
        modInv' = modInv modulus 666666
    f :: Int -> Int -> Int
    f x y
        | (x + y) `mod` 3 /= 0 || 2 * x < y || 2 * y < x
          = 0
        | otherwise
          = modComb (u + v) (u `min` v)
      where
        u = (2 * y - x) `div` 3 :: Int
        v = (2 * x - y) `div` 3 :: Int
  [x, y] <- map unsafeTextToInt . T.words <$> T.getLine :: IO [Int]
  print $ f x y

unsafeTextToInt :: T.Text -> Int
unsafeTextToInt s = case T.signed T.decimal s of
  Right (n, _) -> n
{-# INLINE unsafeTextToInt #-}